

Demonstrating Portfolio Frontier Formation and the Capital Allocation Line Using Interactive Graphing

Pamela LaBorde and David Rystrom

This paper shows how to use interactive graphing via scroll bars to create a two-security portfolio frontier and the Capital Allocation Line. The frontier is enhanced via the use of spinners to control the correlation coefficient and the weight of risky assets as well as data labels that track the weight of the stocks along the frontier as a scroll bar is incremented. The VLOOKUP function is incorporated to lookup the calculated expected return and standard deviation of the risky portfolio at a stock weight controlled by a spinner. The purpose of this paper is to aid instructors in explaining the concepts of investment opportunity sets or frontiers, including the efficient and inefficient portions and riskless lending and borrowing to students of investing.

INTRODUCTION

This paper presents an Excel (Version 2010) spreadsheet¹ that interactively generates the portfolio frontier and the Capital Allocation Line (CAL) through the use of graphing, scroll bars and spinners and the VLOOKUP function. This technique allows the instructor to demonstrate: (1) the effect of changing the weight of securities in the portfolio; (2) the effect of the correlation coefficient on the position and shape of the frontier; and (3) the impact of incorporating a riskless asset upon the frontier. We have found this to be an excellent method of teaching these concepts without becoming mired in calculations and tediously redrawing the graphs by hand, and more effective than showing students static graphs from textbooks.

The paper proceeds as follows. The first section reviews the basic idea of portfolio theory and the "traditional" method of explanation and demonstration. The second section explains the process of creating the graph with the frontier of two risky assets, then adding the data labels and the spinner that controls the correlation coefficient. The final section incorporates a riskless asset and generates the CAL via scroll bars that control the borrowing and lending weights and includes a spinner to control the allocation of risky funds between two risky assets.

TRADITIONAL METHOD AND PRELIMINARY DISCUSSION

The two-security portfolio is the starting point for discussion of the frontier and the portfolio choice problem. The Expected Return (ER) of a two-security portfolio is:

$$\text{Expected Return}_{\text{portfolio}} = (w_1 \times \text{ER}_1) + (w_2 \times \text{ER}_2) \quad (1)$$

and the equation for the Standard Deviation of the two-security portfolio is:

$$\text{SD}_{\text{portfolio}} = \sqrt{w_1^2 \text{SD}_1^2 + w_2^2 \text{SD}_2^2 + 2w_1 w_2 \text{SD}_1 \text{SD}_2 \rho_{1,2}} \quad (2)$$

where:

w_1 = weight in security 1,
 w_2 = weight in security 2,
 ER_1 = expected return (ER) for security 1,
 ER_2 = expected return (ER) for security 2,
 SD_1 = standard deviation (SD) for security 1,
 SD_2 = standard deviation (SD) for security 2, and
 $\rho_{1,2}$ = correlation coefficient between securities 1 and 2.

Traditionally, a textbook (e.g., Heath and Zaima [2006] or Bodie, Kane, and Marcus [2005]) will present these equations, enter some typical values and illustrate the calculations. Then the weights and correlation coefficient are varied to show the investment opportunity set and to explore the effect of different values upon the shape of the frontier. In classroom presentations, this may involve tedious calculations via the board or, alternatively, presentations via PowerPoint or overhead projections in which the results are already pre-drawn; these approaches may not leave the student with a clear understating of how the various points on the frontier are generated. Our interactive approach can be used to show immediately and clearly how changing inputs impact the position and shape of the frontier.

THE INTERACTIVE SPREADSHEET

Inputs

We'll begin with an empty spreadsheet tab. The inputs needed for the basic portion of the spreadsheet are the expected return (ER) and standard deviation (SD) for each of the two securities (STOCK1 and STOCK2) and the correlation coefficient ($\rho_{1,2}$) for the two securities. For later use we also include the return on a riskless asset. These data are all entered by the instructor--no calculations are involved. The basic inputs for this example are shown in Table 1. Input the values shown in Table 1 into the spreadsheet.

Table 1. Basic Inputs.

	A	B	C	D	E
1		Stock1	Stock2	Riskless Asset	Correlation Coefficient
2	Expected Return (ER)	22.20%	13.80%	3%	-0.46
3	Standard Deviation (SD)	12.11%	20.72%	0%	

Next, add the possible weights of STOCK1 in cells B7:B107 and labels as shown in B5:D6 as shown in Table 2A and 2B.

Table 2A. Weight, SD and ER for Risky Portfolio (Values).

	B	C	D
5		Risky Portfolio	
6	Weight in Stock1	SD	ER
7	0%	20.72%	13.80%
8	1%	20.46%	13.88%
9	2%	20.20%	13.97%
...			
105	98%	11.68%	22.03%
106	99%	11.90%	22.12%
107	100%	12.11%	22.20%

Table 2B. Weight, SD and ER for Risky Portfolio (Formulas)

	B	C	D
5		Risky Portfolio	
6	= "Weight in "&B1	SD	ER
7	0	=SQRT((B7^2)*(\$B\$3^2)+((1-B7)^2*(C\$3^2)+(2*B7*(1-B7)*\$B\$3*C\$3*\$E\$2)))	=(B7*\$B\$2)+((1-B7)*(C\$2))
8	0.01	=SQRT((B8^2)*(\$B\$3^2)+((1-B8)^2*(C\$3^2)+(2*B8*(1-B8)*\$B\$3*C\$3*\$E\$2)))	=(B8*\$B\$2)+((1-B8)*(C\$2))
9	0.02	=SQRT((B9^2)*(\$B\$3^2)+((1-B9)^2*(C\$3^2)+(2*B9*(1-B9)*\$B\$3*C\$3*\$E\$2)))	=(B9*\$B\$2)+((1-B9)*(C\$2))
...			
104	0.97	=SQRT((B104^2)*(\$B\$3^2)+((1-B104)^2*(C\$3^2)+(2*B104*(1-B104)*\$B\$3*C\$3*\$E\$2)))	=(B104*\$B\$2)+((1-B104)*(C\$2))
105	0.98	=SQRT((B105^2)*(\$B\$3^2)+((1-B105)^2*(C\$3^2)+(2*B105*(1-B105)*\$B\$3*C\$3*\$E\$2)))	=(B105*\$B\$2)+((1-B105)*(C\$2))
106	0.99	=SQRT((B106^2)*(\$B\$3^2)+((1-B106)^2*(C\$3^2)+(2*B106*(1-B106)*\$B\$3*C\$3*\$E\$2)))	=(B106*\$B\$2)+((1-B106)*(C\$2))
107	1	=SQRT((B107^2)*(\$B\$3^2)+((1-B107)^2*(C\$3^2)+(2*B107*(1-B107)*\$B\$3*C\$3*\$E\$2)))	=(B107*\$B\$2)+((1-B107)*(C\$2))

1. Enter the labels as shown in B5:D6.

2. Type "0%" in B7 and "1%" in B8. (NOTE ON QUOTATION MARKS: When a value or phrase is highlighted with quotation marks, do NOT input the quotation marks--the marks are merely to separate the information from the surrounding text. However, if quotation marks appear within a formula, the quotation marks should be included in the formula inputted.) Select the range B7:B8 with the mouse and use the "fill handle" (the small black square at the bottom right of the active cell or range) to drag the formulas to the range B9:B107.

3. Place the formula for Standard Deviation in C7:

$$=SQRT((B7^2)*(B3^2)+((1-B7)^2*(C$3^2)+(2*B7*(1-B7)*$B$3*C$3*E2)))$$

C7 should now contain "0.2072," the SD of a portfolio composed of 0% in STOCK1 and 100% in STOCK2.

4. Place the formula for Expected Return in D7:

$$=(B7*B2)+((1-B7)*(C$2))$$

D7 should now contain the value "0.138," the ER of the portfolio. A useful exercise at this point is to ask the students to help generate the equations in Steps 3 and 4 by asking which cells should be referenced and if the cell references should be absolute or relative. This exercise offers a good opportunity to explain to students the difference between absolute and relative references and when each is applicable to a particular situation. Absolute references are used for the ER and SD of the two stocks as well as the correlation coefficient.

5. Select C7:D7 and use the fill handle to drag the formulas to C8:D107. Range C7:D107 now contains the


expected return and standard deviation of the portfolio starting with 0% in STOCK1 and 100% in STOCK2 in increments of 1%.

The formulas for the values obtained in Table 2A are shown in Table 2B. It is important to note that the ONLY cells created during the lecture are B7:D107. The remainder of the spreadsheet discussed in the paper is calculated before the lecture. Now is a good time to perform several "housekeeping" functions. First, rename the worksheet tab to "Data" from "Sheet1." Secondly, save the spreadsheet.

Graphing the Frontier

While one could scroll down the list of values calculated in C7:D107 and discuss how risk and return changes at various weights of STOCK1 and STOCK2, showing the portfolio's ER and SD graphically helps students to intuitively grasp the issue of efficient vs. inefficient weightings. One could graph cells C7:D107 and show the entire frontier onscreen at once. However, there is some benefit to starting with an "empty" graph and using a scroll bar to draw the frontier. The chart we'll be creating for this section is shown in Figure 1. It is an XY Scatter Plot of the ER and SD of a portfolio composed of STOCK1 and STOCK2, with the weights changing in 1% increments via the "Frontier Scroll Bar" and a spinner controlling the value of the correlation coefficient between STOCK1 and STOCK2.

Figure 1. Graph with Frontier Scroll Bar and Correlation Coefficient Spinner



To create the graph and scroll bar:

1. Input the following formula in J7:

`=IF(K5/100>=$B7,C7,NA())`

This formula checks the value of the counter (K5--which is currently empty but will be linked to a value created by the scroll bar), converts the value to a decimal, and compares this decimal value to the value stored in B7. B7 contains the weight of STOCK1. If the value of the counter is equal to or greater than the weight of STOCK1, the SD as calculated in C7 is entered into J7; otherwise the value NA() is entered. NA() returns the error value #N/A which means that no value is available. When a cell in a graphed data series contains the #N/A value, nothing appears on the graph for that point.

2. Using the fill handle, drag the formula in J7 to J8:J107. J7 should now contain the SD of the portfolio while J8:J107 should contain #N/A as K5 is currently empty. To test whether or not the formula is working, input a value of "3" into K5. The values of J8:J10 should change from #N/A to the SDs shown in C8:C10.

3. Copy the formulas in J7:J107 to K7:K107. K7:K10 should contain the ERs of the portfolio as shown in D8:D10

while K11:K107 should contain #N/A. Panel A of Table 3 shows the values while Panel B shows the formulas for a select range of this section of the spreadsheet.

4. Create an XY Scatter Plot similar to the one shown in Figure 1 by selecting J7:K107, choose the Insert ribbon, and selecting "Scatter" from the "Charts" section. Choose the "Scatter with only markers" chart option. Excel will automatically place J7:J107 as X and K7:K107 as Y. Four points should appear in the Plot Area of the graph as we've entered a test value of "3" in K5 and there are four XY data points in J7:K107. To avoid re-scaling of the X and Y axes as the scroll bar is scrolled, set the X axis scale to 25% by selecting the chart, then Chart Tools/Layout/Axes/Primary Horizontal Axis/More Primary Horizontal Axis Options and set the Axis Options maximum value to a "fixed" value of "0.25" and minimum value to "0." Do the same for the Y axis scale using the Primary Vertical Axis choice.

5. "Turn off" the legend and add axis labels and a chart title as shown in Figure 1 using the features under Chart Tools/Layout.

6. Finally, we prefer to place the chart in its own worksheet tab. To do this, right click in the "Chart Area"

(the area of the chart that does not include the actual plot of data points) of the newly-created chart, choose "Move Chart" and input "Frontier" in the "New Sheet" input box.

7. Resize the "Plot Area" by clicking the "Plot Area" and dragging the handle in the upper right corner so that the plot area is off-center with space above and to the right for the scroll bars and spinners.

Table 3. ER and SD For XY Scatter Plot (Frontier). Panel A: Values

	J	K
5		3
6	Risky SD	Risky ER
7	20.72%	13.80%
8	20.46%	13.88%
9	20.20%	13.97%
10	19.93%	14.05%
11	#N/A	#N/A
12	#N/A	#N/A
...		
	J	K
105	#N/A	#N/A
106	#N/A	#N/A
107	#N/A	#N/A

Panel B: Formulas


	J	K
5		3
6	Risky SD	Risky ER
7	=IF(\$K\$5/100>=\$B7,C7,NA())	=IF(\$K\$5/100>=\$B7,D7,NA())
8	=IF(\$K\$5/100>=\$B8,C8,NA())	=IF(\$K\$5/100>=\$B8,D8,NA())
9	=IF(\$K\$5/100>=\$B9,C9,NA())	=IF(\$K\$5/100>=\$B9,D9,NA())
10	=IF(\$K\$5/100>=\$B10,C10,NA())	=IF(\$K\$5/100>=\$B10,D10,NA())
11	=IF(\$K\$5/100>=\$B11,C11,NA())	=IF(\$K\$5/100>=\$B11,D11,NA())
12	=IF(\$K\$5/100>=\$B12,C12,NA())	=IF(\$K\$5/100>=\$B12,D12,NA())
...		
	J	K
105	=IF(\$K\$5/100>=\$B105,C105,NA())	=IF(\$K\$5/100>=\$B105,D105,NA())
106	=IF(\$K\$5/100>=\$B106,C106,NA())	=IF(\$K\$5/100>=\$B106,D106,NA())
107	=IF(\$K\$5/100>=\$B107,C107,NA())	=IF(\$K\$5/100>=\$B107,D107,NA())

Adding the "Frontier Scroll Bar" to the graph requires accessing the Developer ribbon. If the Developer ribbon is not currently available, activate it by choosing File/Options/Customize Ribbon and check the box on the right side of the dialog box next to "Developer" under "Main Tabs." Click OK to return to your spreadsheet. To add the scroll bar:

1. Select the chart. On the Developer ribbon, choose Insert then choose "Scroll Bar (Form Control)" from the "Form Controls." Use the mouse to position the scroll bar on the spreadsheet. Right click the scroll bar and select "Format Control." Select the "Control" tab if it is not automatically chosen. The Control tab of the Format Control dialog box is shown in Figure 2.

The values to be entered are minimum value, maximum value, incremental change and cell link. The minimum value should be 0 (the lowest weight that can be placed into

Figure 2. Control Tab of the "Frontier Scroll Bar."



STOCK1), the maximum value should be 100 (the highest weight that can be placed into STOCK1) and the incremental value is the amount by which the counter will change when the scroll bar is scrolled with the arrows on either end--we use 1, or a 1% change in weight--you might wish to enter 5 or 10 so that the frontier is drawn more quickly. Cell Link refers to the sheet and cell in which you wish the counter to be placed. For this spreadsheet "Data" is the sheet name and \$K\$5 is the cell within the "Data" sheet.

2. Place your cursor in the "Cell Link" input box then click on cell K5 in your Data worksheet to link the scroll bar. Input the appropriate minimum, maximum and incremental values in the boxes, select OK and the scroll bar is ready for use. As the "Frontier Scroll Bar" is scrolled to the right, the counter increases in value in increments of 1 (or 1%) resulting in increasing values in K5; and the values in J8:K107 change from #N/A to the SD and ER of the portfolio. Since the frontier on the graph is created from cells J7:K107 (and not from C7:D107), as these cells change from #N/A to the values for SD and ER, the frontier is drawn.

3. Finally, add a label to the scroll bar by selecting the Frontier worksheet, selecting the chart, and choosing "Shapes" from the Insert ribbon. Choose "Pentagon" from the "Block Arrows" section. Place and size the arrow with the mouse. Right click the arrow and choose "Format Object" to adjust the colors, outlines, etc. Lastly, right click the arrow, select "Edit Text" and type "Frontier Scroll Bar" in the arrow.

For class presentation, the chart is already developed when demonstrating the spreadsheet to students. The chart is on a separate worksheet when first accessing the spreadsheet during lecture. The "Frontier Scroll Bar" is positioned at the minimum value when class begins. It is then possible to use the scroll bar on the frontier sheet to show the effect of changing the weights. As the scroll bar is

incremented to the right, K5 increases in value (to a maximum of 100, or 100% weight in STOCK1).

Enhancing the Display: Use of Data Labels

To provide more information on the interactive chart, data labels can be used. For example, one could choose to display the weight of each stock based upon the current value of the scroll bar. It is also helpful to have the endpoints for the frontier labeled with the security names or ticker symbols such as those in B1 and C1.

To add the two endpoints:

1. Right click on the Chart Area and choose "Select Data." Choose "Add" in the "Legend Entries (Series)" dialog box. The cursor will be in the input box for "Series Name."

2. Select B1 (which contains the ticker symbol "STOCK1") from the Data worksheet. Next, place the cursor in the input box for "Series X values" and click on B3 in the Data worksheet. Data!B3 contains 12.11%. This is the SD for STOCK1.

3. Select the input box for "Series Y values" and delete the "{1}" (keep the "=" sign). Click on B2 from the Data worksheet. Data!B2 contains the ER for STOCK1, or 22.20%.

4. Choose "OK" and "OK" to close the dialog boxes and a new data point should appear on the graph. Move the slider for the Frontier Scroll Bar to the far left to make it easier to see the newly-created data point. Right click the data point and choose "Add Data Labels" then right click the data point again and choose "Format Data Labels." Under "Label Options" check "Series Name" for "Label Contains" and uncheck "Y Value." Close the dialog box and the data point should now be labeled "STOCK1."

5. If desired, you can change the marker options so that the data point blends with the frontier. To do so, right click the data point again and choose "Format Data Series." Select "Marker Options", set the "Marker Type" to "Built-in" and choose the circle. Change the color to blue, by using the "Marker Fill/Solid Fill" controls.

6. Repeat steps 1 through 5 for the STOCK2 data point, using "Data!C1" as the "Series Name," "Data!C3" as the "Series X values" and "Data!C2" as the "Series Y values."

Next, add the data labels that track the weight of STOCK1 and STOCK2 based upon the current value of the scroll bar as it is scrolled from the minimum to the maximum value. Although this section is a bit tedious, the results greatly enhance both the graph and student comprehension. Create this data label by inputting the following equation in L5 on the Data worksheet.

```
=CONCATENATE(K5,"% in ",B1,",",
(1-K5/100)*100,"% in ",C1)
```

The CONCATENATE() function joins a group of characters and values into a single text string, which we can use as a data label. Assuming your scroll bar is set at the minimum value, L5 should contain "0% in STOCK1, 100% in STOCK2." As you move the scroll bar the value in K5 will change along with the data label in L5.

The data label in L5 will be used as a "Series Name" for a new series on the chart. To create this series we need to track whether or not a point is being displayed in range J7:K107. It is best to only display a label for the newest data point added by the scroll bar. Thus, if a point is not displayed, there should be no display of a data label. As newer data points are added via the scroll bar, this method provides the ability to "turn off" data labels on the points already graphed and "turn on" the data label for the newest point. This can be accomplished by activating a specific data point, graphing it and assigning to that specific point the data label just created in L5. The process requires nested AND() and ISNA() statements within an IF() statement.

L7:L107 will track whether a data label should be "on" or "off" for a data point.

1. Enter the following formula in L7 and using the fill handle drag the formula to L8:L107.

```
=IF(AND(ISNA(K7)=FALSE,ISNA(K8)=TRUE),K7,
NA())
```

2. Set the scroll bar to a test value of 10 either by scrolling or typing "10" in K5. At this point, L17 should contain a value of "0.1464"--the ER of the portfolio composed of 10% in STOCK1 and 90% in STOCK2. All the remaining cells in L7:L107 should contain "#N/A." The ISNA() function returns a value of "TRUE" if a cell contains "#N/A" and "FALSE" if a cell does not contain "#N/A." The IF statement uses the following syntax: =IF(logical_test, [value_if_true], [value_if_false]). Do not include the brackets "[" or "]" Evaluate the formula for L17. L17 contains the following:

```
=IF(AND(ISNA(K17)= FALSE, ISNA(K18)= TRUE),
K17,NA())
```

The logical_test [AND(ISNA(K17)= FALSE, ISNA(K18)= TRUE)] requires K17 to have a viable data point (i.e., is not equal to #N/A) and the next cell (K18) to not have a data point (i.e., is equal to #N/A). This makes K17 the newest data point created by the scroll bar. If both conditions are satisfied, the [value_if_true] activates and turns on the data point in L17 creating a point that can be graphed with the data label in L5. If either of the logical tests fail, the [value_if_false] activates and places #N/A in the cell. This situation occurred in L18, for instance, and all the remaining cells in the range L7:L107. Test the formulas in L7:L107 by typing in various values (weights) in K5 (between 0 and 100). The values in L7:L107 should remain #N/A except for the row that corresponds to the weight input in K5. Thus, if K5 contains the value "23," L30 should contain the value "0.15732" but the remaining values in L7:L107 should be #N/A. As the Frontier Scroll Bar is scrolled, L7:L107 will contain the ER for only the one data point we want to label. The SD for that data point is located in column J.

3. Add L7:L107 to the chart by right clicking on the chart, choosing "Select Data" and "Add." The "Series Name" is the data label in L5, or "=Data!L5", the "Series X Values" are "Data!J7:J107" and the "Series Y Values" are "Data!L7:L107". Select "OK" and "OK" to exit the dialog boxes. Assuming a value in K5 of "23," the chart should

contain a new data point at a SD of 0.15058 and an ER of 0.15732. It may be difficult to see the new data point as the point will be atop the last point generated by the Frontier Scroll Bar.

4. Right click this new data point, and choose "Add Data Labels." The data point should now have the label "0.15732." Right click the data point, choose "Format Data Labels" and in the "Label contains" section check the box for "Series Name" and uncheck the box for "Y value." Set the "Label Position" to "Left" and Close the dialog box. While moving the scroll bar, the data point should track the ER and SD for the portfolio given the current weight of STOCK1 as determined by the scroll bar. The data label should change to display the weights that determine that data point.

The result is a dynamic chart--the values will automatically adjust as changes are made in the basic inputs located in B1:C3. The interactive nature of the chart grabs the students' attention and the changing data labels showing the weights of STOCK1 and STOCK2 make it clear to the students exactly what is occurring as the frontier is created via the Frontier Scroll Bar.

Pedagogic Use of the Interactive Feature

Classroom discussion can utilize the interactive feature to discuss what portion of the frontier is efficient vs. inefficient. This can be shown by moving the scroll bar to show different points on the opportunity set. The students observe the greater ERs on the higher, efficient part of the frontier, as compared to the lower inefficient corresponding points for any SD. At the same time, the data label displays the weights that produce those combinations. Students can easily follow and understand the idea of dominant portfolios and the efficient part of the opportunity set from this demonstration. Additionally, students can observe what weights in STOCK1 and STOCK2 cause the frontier to switch from the efficient portion to the inefficient portion--the minimum standard deviation portfolio.

The instructor can also effectively use the interactive feature to demonstrate the impact of a change in the correlation coefficient upon the frontier. We'll add a spinner control to the chart that determines the value of the correlation coefficient in Data!E2. This should give the student a clear visual sense of the impact of the correlation coefficient on the shape of the frontier.

Spinner Control for Correlation Coefficient

Since the chart and the data are on two separate worksheets, we add a spinner control to the chart along with a text box that shows the current value of the correlation coefficient to the graph.

1. In the Frontier worksheet, choose "Insert" from the Developer ribbon and select "Spin Button (Form Control)." With the mouse, place and size the spin button in the upper left of the chart. Right click the spin button (or spinner) and select "Format Control." The minimum and maximum

values need to represent the possible range for the correlation coefficient, or -1 to +1. Unfortunately, minimum value does not allow negative numbers nor is the incremental change allowed to be less than 1. Therefore, set the minimum value to "0," the maximum value to "200," and the incremental change value to "10." Finally, set the cell link to "Data!F2."

2. The spinner places a value ranging from 0 to 200 in Data!F2. However, correlation coefficients can only be in the range of -1 to +1. To convert the spinner range to that of the correlation coefficient, we can divide the spinner value by 100 (giving us a minimum value of 0 and a maximum value of 2). By subtracting 1 from the result, the range of the correlation coefficient, the desired range extending from a minimum value of -1 and a maximum value of +1, is created. Additionally, it is necessary to replace the original correlation coefficient value in cell E2 with the converted value generated by the spinner. To do so, on the Data worksheet enter the following equation in E2:

$$=F2/100-1$$

Set the scroll bar to its maximum position and note how the shape of the frontier changes as you adjust the spinner. To display the value of the correlation coefficient, we'll add a text box on the graph.

3. On the Frontier worksheet, select "Text Box" from the Insert ribbon. Using the mouse, draw a text box next to the spinner. Place your cursor in the formula bar (the input area to the right of the f_x) and type "Data!E2." The current value for the correlation coefficient should appear in the text box. Note that if you enter the formula into the textbox itself and not via the formula bar, the text box will display "Data!E2" and not the actual value. Give the spinner a whirl to ensure the value in the text box adjusts with the spinner. You may wish to add a second text box with the text "Correlation Coefficient" as shown in Figure 1.

ADDING THE RISKLESS ASSET AND THE GENERATION OF THE CAPITAL ALLOCATION LINE

After the students are familiar with and understand the development of the frontier and the efficient/inefficient portions of the frontier, it is possible to introduce the idea of combining risky portfolios on the frontier with the riskless asset. These combinations are sometimes called the Capital Allocation Line (CAL)², the term used in this paper. It is useful to begin with a risky portfolio on the frontier that is not the tangency point for the CAL. This example uses the risky portfolio that is equally-weighted between STOCK1 and STOCK2. Typically an equally-weighted portfolio is relatively easily envisioned by students. This portfolio is denoted as risky portfolio E as shown in Figure 4.

Next, form a portfolio of 70% of our funds in portfolio E and 30% in the risk-free asset as another example. The use of dollar amounts can help clarify the weighting. As an example, begin with an investment of \$100,000, and ask what ER and SD will result if \$30,000 is invested in the riskless asset and \$70,000 in portfolio E. If the riskless

asset generates a return of 3% and risky portfolio E has an ER of 18% and a SD of 9.29% (assuming a correlation coefficient of -0.46 between STOCK1 and STOCK2), then the portfolio composed of 30% in the riskless asset and 70% in portfolio E will have an ER of 13.5% and a SD of 6.5% as demonstrated below:

$$ER_{E, \text{risk-free}} = (0.3)(0.03) + (0.7)(0.18) = 13.5\% \quad (3)$$

$$SD_{E, \text{risk-free}} = [(0.3^2)(0.0^2) + (0.7^2)(0.0929^2) + 2(0.3)(0.7)(0)(0.0929)(0)]^{1/2} = [0 + (0.7^2)(0.0929^2) + 0]^{1/2} = (0.7)(0.0929) = 6.5\% \quad (4)$$

It is helpful to work this out "by hand" and discuss where this point would plot on the Frontier graph.

After illustrating these ideas with two different portfolios, the next step is to show students via the graph in Figure 4 how combining various weights (both positive and negative) of the riskless asset with risky portfolio E results in a line connecting the risk-free return and risky portfolio E. The next section shows how to add the Lending CAL and Borrowing CAL scroll bars to the graph.

Basic inputs for the CAL

The process for generating these scroll bars is similar to that used for the "Frontier Scroll Bar." To generate the Lending CAL, on the Data worksheet:

1. Enter the risk-free rate (3%) in D2 and the SD of the riskless asset (0%) in D3 if not already entered.
2. Create a section of the spreadsheet that contains the ER and SD for the risky portion of the portfolio--portfolio E in Figure 4. The SD and ER have already been calculated in columns C and D; it is merely a matter of having Excel find the values for the specific weight of STOCK1 and STOCK2 that create risky portfolio E.

Panel A of Table 4 shows the section of the spreadsheet used to extract the necessary data while Panel B contains the formulas. P4 contains the weight of STOCK1 for risky portfolio E and is a value determined by the user of the spreadsheet. For the example in Equations 3 and 4, P4 would have a value of 50%, or 0.50 as the risky portfolio E is comprised of equal weights in STOCK1 and STOCK2.

Table 4. Basic Inputs for CAL.

Panel A: Values

	P	Q	R	S
3	Weight STOCK1	Weight STOCK2	Risky ER	Risky SD
4	50%	50%	18.00%	9.29%

Panel B: Formulas

	P	Q	R	S
3	Weight STOCK1	Weight STOCK2	Risky ER	Risky SD
4	0.5	=1-P4	=VLOOKUP(\$P\$4,\$B\$7:\$D\$107,3)	=VLOOKUP(\$P\$4,\$B\$7:\$D\$107,2)

1. Add labels in P3:S3 as shown in Table 4 (Panel A).
2. Enter a value of "0.5" in P4. Q4 is calculated as (1 – P4) or 1 minus the weight of STOCK1. Enter "=1-P4" in Q4. Next, use the VLOOKUP formula to find the corresponding ER and SD for the risky portfolio comprised of P4 weight of STOCK1 and Q4 weight of STOCK2.

3. Enter the following formula in R4:
=VLOOKUP(\$P\$4,\$B\$7:\$D\$107,3)

The VLOOKUP (Vertical LOOKUP) formula uses the following convention: VLOOKUP (lookup_value, table_array, col_index_num). The lookup_value is the weight in P4. Excel will attempt to find that value in the table_array, or the range \$B\$7:\$D\$107. If it finds the value contained in P4 in the first column of \$B\$7:\$D\$107, it will return as an answer the value in the third (3) column, or col_index_num, of the corresponding row. R4 should contain the ER for a portfolio weighted 50% in STOCK1 and 50% in STOCK2, or "0.18."

4. Enter the following formula in S4:
=VLOOKUP(\$P\$4,\$B\$7:\$D\$107,2)

The formula performs the same function as the one in R4 except it returns the SD from the second column in the range of B7 to D107. Cells R4 and S4 now contain the ER and SD of risky portfolio E for generating the CAL. If you wish to evaluate the effect of different weights of STOCK1 and STOCK2 for the risky portfolio E, you only need to change the value in P4. Excel will look up the SD and ER of the new risky portfolio E and place the results in R4 and S4. Next, add a point for risky portfolio E to the graph.

5. From the Frontier worksheet, right click the Chart Area and select "Select Data" then "Add." Input "E" as the "Series Name," "Data!S4" as the "Series X values" and "Data!R4" as the "Series Y values." Select OK and OK to close the dialog boxes. While it may be difficult to see, a new data point is now on the graph--at the bottommost point where the CAL intersects the frontier. It is easier to find the new data point if the Frontier Scroll Bar is in the minimum position.

6. Right click this new point and select "Add Data Labels." The label "0.18" should appear after you close the dialog box. Right click the new point again, select "Format Data Labels," check the box by "Series Name," uncheck the "Y Value" box and check the "X Value" box. Also, change the "Label Position" to "Above." Select "Close" to exit the dialog box. The result is a data label with point E and the accompanying SD.

7. Finally, right click the point one final time, select "Format Data Series," choose the square "Built-in" Marker

Option and set the "Marker Fill" to a "Solid Fill" in a bright color.

Next, generate the ER and SD data for combined risky portfolio E and the risk-free asset, then generate another two columns that contain the data points to be graphed for the Lending CAL as shown in Panel A of Table 5. Range H7:I107 on the Data worksheet contains the information used to create the data points to be graphed as the "Lending CAL" scroll bar is scrolled in the same way J7:J107 generates the frontier as the Frontier Scroll Bar is scrolled. Initially H7:I107 contain the value #N/A, but the values change to the SD and ER as the Lending CAL scroll bar is scrolled. Panel B of Table 5 shows the formulas used to generate these data.

E7:E107 contains the weights for the risky portfolio E, while F7:G107 contain the SD and ER for the combined risky portfolio E and the riskless asset. The formulas for SD and ER in F7:G107 obtain the values for the risky portfolio portion with the weight combination established by the value in P4 and the SD (S4) and ER (R4) for that combination of risky assets.

Incorporate the information from Table 5 into your spreadsheet:

1. Enter the labels in E6:I6 into the Data worksheet.
2. Input the weights of "0%" and "1%" into E7 and E8, respectively, then highlight E7:E8 with the mouse and use the fill handle to drag the weights into E9:E107.

Table 5. Inputs for the Lending CAL Scroll Bar. Panel A: Values

	E	F	G	H	I
6	Weight (Risky)	SD	ER	CAL Graph Inputs	
7	0%	0.00%	3.00%	0.00%	3.00%
8	1%	0.09%	3.15%	#N/A	#N/A
9	2%	0.19%	3.30%	#N/A	#N/A
...					
105	98%	9.10%	17.70%	#N/A	#N/A
106	99%	9.20%	17.85%	#N/A	#N/A
107	100%	9.29%	18.00%	#N/A	#N/A

Table 5. Inputs for the Lending CAL Scroll Bar. Panel B: Formulas

	E	F	G	H	I
6	Weight (Risky)	SD	ER	CAL Graph Inputs	
7	0	=E7*\$\$S\$4	=\$((1-E7)*\$D\$2)+E7*\$R\$4	=IF(\$N\$7/100>=E7,F7,NA())	=IF(\$N\$7/100>=E7,G7,NA())
8	0.01	=E8*\$\$S\$4	=\$((1-E8)*\$D\$2)+E8*\$R\$4	=IF(\$N\$7/100>=E8,F8,NA())	=IF(\$N\$7/100>=E8,G8,NA())
9	0.02	=E9*\$\$S\$4	=\$((1-E9)*\$D\$2)+E9*\$R\$4	=IF(\$N\$7/100>=E9,F9,NA())	=IF(\$N\$7/100>=E9,G9,NA())
...					
105	0.98	=E105*\$\$S\$4	=\$((1-E105)*\$D\$2)+E105*\$R\$4	=IF(\$N\$7/100>=E105,F105,NA())	=IF(\$N\$7/100>=E105,G105,NA())
106	0.99	=E106*\$\$S\$4	=\$((1-E106)*\$D\$2)+E106*\$R\$4	=IF(\$N\$7/100>=E106,F106,NA())	=IF(\$N\$7/100>=E106,G106,NA())
107	1	=E107*\$\$S\$4	=\$((1-E107)*\$D\$2)+E107*\$R\$4	=IF(\$N\$7/100>=E107,F107,NA())	=IF(\$N\$7/100>=E107,G107,NA())

3. Input the following equation into F7:
=E7*\$\$S\$4

This equation calculates the SD for the combination of risky portfolio E and the riskless asset.

4. Input the following equation into G7:
=\$((1-E7)*\$D\$2)+E7*\$R\$4

This equation calculates the ER for the combination of risky portfolio E and the riskless asset.

5. Select F7:G7 with the mouse and use the fill handle to drag the equations to F8:G107.

Just as the Frontier Scroll Bar incremented a counter that was placed in cell K5 the Lending CAL Scroll Bar needs a counter cell reference as well. We'll use N7 as the Lending CAL Scroll Bar's counter cell. This cell will reflect the value of the Lending CAL Scroll Bar and will change the weight of the investment in risky portfolio E. For now, simply type in a value of "10" in N7--this represents a 10% weight in risky portfolio E and, therefore, a 90% weight in the riskless asset. Later we will replace the "10" in N7 with a value generated by the Lending CAL Scroll Bar. First, we need to "turn on" the data points for the Lending CAL for the graph.

1. Input the following formula in H7:
=IF(\$N\$7/100>=E7,F7,NA())

This formula checks the value of the counter (N7), converts it to a decimal, and compares it to the value in E7, which contains the weight of risky portfolio E. If the value of the counter is equal to or greater than the weight of risky portfolio E, the SD as calculated in F7 is entered in H7 (and subsequently will appear on the graph); otherwise the value NA() is entered (and will NOT appear on the graph).

2. The ER for the Lending CAL is determined in the same manner as the SD explained in the previous step. Input the following formula in I7:
=IF(\$N\$7/100>=E7,G7,NA())


This formula will populate the ERs on the graph for the risky portfolio E and the riskless asset.

3. Select the range H7:I7 and drag the formulas to H8:I107. Since the counter cell N7 is set to a weight of "10," H7:I17 should contain the SD and ER from F7:G17, while range H18:I107 should contain #N/A.

Now it is time to add the Lending CAL Scroll Bar to the graph and attach the value of the scroll bar to the counter cell N7.

1. While in the Frontier worksheet, access the Developer ribbon, select "Insert" and choose "Scroll Bar (Form Control)." See Figure 3 for a screenshot of this procedure.

Figure 3. Insert Menu of Developer Ribbon.



With the mouse, position and size the scroll bar vertically on the right side of the plot area of the graph, as shown in Figure 4. Right click the scroll bar and select "Format Control."

2. As with the Frontier Scroll Bar, the values to be entered include the minimum value, maximum value, incremental change and cell link. Input "0" as the minimum


value (the lowest weight that can be placed into risky portfolio E) and "100" as the maximum value (the highest weight that can be placed into risky portfolio E). Input "1" as the incremental change--the amount by which the scroll bar value will change when the arrows on either end are pressed. You may wish to use a larger increment although an incremental value of "1" allows us to fine tune the tangency point weighting scheme of the Lending CAL and the efficient portion of frontier when lecturing on this issue.

3. Link the scroll bar value to N7 on the Data worksheet by inputting "Data!\$N\$7" in the Cell Link section.

4. Select "OK," deselect the scroll bar then scroll the Lending CAL Scroll Bar to test it. As the Lending CAL Scroll Bar is incremented to a higher value, the counter in Data!N7 changes and the values in H7:I107 are changed from #N/A to the SD and ER of the combination of investing in the riskless asset and risky portfolio E.

5. Add an arrow with the text "Lending CAL" to enhance the graph as shown in Figure 4 by choosing the Insert ribbon, then "Shapes" and the "Pentagon" from the "Block Arrows" section. Once you place the pentagon on the chart, select it and rotate it 90 degrees to the right by using the mouse to "grab" the green dot that appears at the top of the pentagon after you select it. Input the appropriate text (right click the pentagon and choose "Edit Text" to add text).

Figure 4. Risky Portfolio Frontier and Capital Allocation Line.



As of yet, the data points in H7:I107 have not been added to the graph. Before we add the data points to the graph, we have chosen to extend the CAL to incorporate riskless borrowing as well as lending. For this spreadsheet we assume a maximum investment in the risky portfolio E of 150% of one's wealth. Thus, we will add another 50 rows to column E on the Data worksheet, extending the weight from 100% to 150%.

1. Select E106:E107 and use the fill handle to drag the formula to E108:E157 to create the weights of 101% to 150%.

2. Highlight F107:G107 and use the fill handle to drag the formula to F108:G157. These cells extend the Lending CAL to the Borrowing CAL.

3. Highlight H107:I107 and use the fill handle to drag the formulas to H108:I108. H108:I108 represent the beginning of the range used to graph the Borrowing CAL. However, we prefer to use a separate scroll bar to graph the borrowing portion of the CAL. This scroll bar will need its own counter cell as N7 is used by the Lending CAL Scroll Bar. Thus, rather than referring to N7 in H108:I108, we need to refer to a new counter cell. We'll use O7 as the counter cell for the Borrowing CAL Scroll bar.

4. Edit the formulas in H108 and I108, changing the "\$N\$7" to "\$O\$7" in both cells.

5. Highlight H108:I108 and use the fill handle to drag the formulas to H109:I157.

To add the Borrowing CAL Scroll Bar to the graph and attach the value of the scroll bar to the counter cell O7:

1. From the Frontier worksheet select the Developer ribbon and "Insert" a "Scroll Bar (Form Control)." Use the mouse to size and position the scroll bar to the right of the Lending CAL Scroll Bar as shown in Figure 4.

2. Right click the scroll bar and select "Format Control."

3. Enter a minimum value of "101" (the minimum weight in the risky portfolio E if one is to be a borrower), a maximum value of "150" (the maximum weight we have chosen for a person borrowing to invest in risky portfolio E), an incremental change of "1" and set the cell link to "Data!O7." 4. Select OK to exit the "Format Control" dialog box.

5. Add an arrow with the text "Borrowing CAL" to enhance the graph as shown in Figure 4 by selecting the Chart Area. From the Insert ribbon, choose "Shapes" and "Pentagon" from the "Block Arrows" section and input the appropriate text (right click the pentagon and choose "Edit Text" to add text). Rotate the pentagon 90 degrees to the right as described earlier.

It is now time to add the data points for the Lending CAL and Borrowing CAL to the graph. Although two separate scroll bars control the lending and borrowing weights of the riskless asset, only one new series of data points is needed on the graph.

1. From the Frontier worksheet, right click in the Chart Area and choose "Select Data." Choose "Add" and input "Data!H7:H157" for the Series X values and "Data!I7:I157" for the Series Y values. Leave the "Series Name" blank.

2. Select "OK" and "OK" to close the dialog boxes. Test the Lending CAL Scroll Bar and the Borrowing CAL Scroll Bar to ensure that the links are functioning properly. Note that it looks awkward having the Borrowing CAL Scroll Bar at any value other than the minimum value if the Lending CAL Scroll Bar is not at its maximum value.

In classroom presentation, we explore different weights for the riskless asset and portfolio E to show different points on the CAL. The instructor can then ask students some leading questions, such as:

- Is there a better point (line) than the 50/50 weighting of STOCK1 and STOCK2?

- If so, how do you get there? [Specifically, should the weight of STOCK1 change (increase or decrease)?]

- Will investing 100% in the highest-return stock accomplish the best point?

By changing the weight of STOCK1 in P4 we can incorporate the students' responses. (NOTE: We perform this portion of the demonstration with all the scroll bars at their maximum values.) However, as P4 is on a separate worksheet from the chart, adding a spinner to the chart to control the weight of STOCK1 in Data!P4 will enable us to easily adjust the value of P4.

Spinner Control for Weight in STOCK1 of Risky Portfolio E

Just as we added a spinner to control the value of the correlation coefficient, adding a spinner to control the risky portfolio E enhances the visual presentation. We'll also add a textbox showing the weight in STOCK1 on the graph.

1. In the Frontier worksheet, choose Insert from the Developer ribbon and select "Spin Button" (Form Control). With the mouse place and size the spin button in the upper right of the chart. Right click the spinner and select "Format Control." The minimum and maximum values need to represent the possible range for the weight in STOCK1, which can range from 0% to 100%. Set the minimum value to "0," the maximum value to "100," the incremental change value to "1," and the cell link to "Data!P5."

2. The spinner places a value ranging from 0 to 100 in Data!P5. However, the weight should be a decimal amount rather than an integer. To convert the spinner range to a decimal, we can divide the spinner value by 100 (giving us a minimum value of 0 and a maximum value of 1). We need to replace the original STOCK1 weight in P4 with the converted value generated by the spinner. To do so, on the Data worksheet enter the following equation in P4:

$$=P5/100$$

Set all the scroll bars to their maximum position and note how the slope of the CAL changes as you adjust the spinner on the graph. Next, we'll add a text box to display the current weight of STOCK1 in risky portfolio E.

3. On the Frontier worksheet, select "Text Box" from the Insert ribbon. Using the mouse, draw a text box next to the spinner. Place your cursor in the formula bar (the area to the right of the f_x) and input "Data!P5." The current value for the spinner should appear in the text box. Add another

text box with the text "Portfolio E: % in STOCK1" as shown in Figure 4.

By interacting with the students as described above, we can lead students to the conclusion that the tangency point on the efficient portion of the frontier will produce an optimal CAL and that there is some optimal weighting of STOCK1 and STOCK2 that will achieve that point. While it is possible to determine the tangency point in Excel via the Solver function and VBA code, we generally have the students determine the tangency point by "eyeballing" it.

SUMMARY

The spreadsheet and interactive graph we present here strike a nice compromise between lengthy numerical calculations and superficial presentation. They allow the instructor to show how the frontier is generated and the effect of changing weights and/or the effect of different correlations. The demonstration also introduces the use of a riskless asset combined with borrowing and lending as a precursor to the idea of the Capital Market Line and capital market theory. We have found that students can grasp these concepts quickly and respond to questions easily when the interactive version is presented to them.

There are some obvious extensions that can enhance this spreadsheet:

1. Perform an interactive web query that will look up the prices for two stocks from some convenient website such as finance.yahoo.com and use that price data to calculate historical returns for estimating the inputs. (See Carter, et al, [2002] for one way to accomplish this).
2. Extend the spreadsheet to more than two stocks.

ENDNOTES

- 1 The paper assumes the reader has a basic understanding of Excel, including the difference between absolute and

relative references/notation, and basic graphing using XY Scatter Plots.

2 The Capital Allocation Line connects any point on the frontier with the riskless asset. If the frontier were generated from all risky assets, the line extending from the riskless asset to the tangency point on the efficient portion of the frontier would be the well-known Capital Market Line.

REFERENCES

- Bodie, Z., A. Kane, & J. Marcus (2005). *Investments*. New York: McGraw Hill/Irwin.
- Carter, D., W. Dare & W. Elliott (2002). Determination of Mean-Variance Efficient Portfolios Using an Electronic Spreadsheet. *Journal of Financial Education*, 28, 63-78.
- Hearth, D., & J. Zaima (2006). *Contemporary Investments: Security and Portfolio Analysis*. New York: Southwestern.
- Markowitz, H. (1952). Portfolio Selection. *Journal of Finance*, 5, 77-91.
- Stephens, A. (1998). Markowitz and the Spreadsheet. *Journal of Financial Education*, 24, 35-43.
- Walkenbach, J., (2010). *Excel 2010: Power Programming with VBA*. Indianapolis: Wiley Publishing Inc.

Pamela LaBorde is an Assistant Professor at Western Washington University. Her current research interests include pedagogical issues and personal finance.

David Rystrom is an Associate Professor of Finance at Western Washington University. He received his Ph. D. at the University of Oregon.